
irstFew is a short Wimp compliant Basic
application that allows you to have a

look at the first 217 bytes of any file
dragged to it. This can be very handy, for
example, when trying to sort out the latest
versions of various programs or modules,
as the version identification is usually given
in the first couple of lines of code. Thus, to
check which version of CLib you have, just
drag the CLib module to the FirstFew icon,

and a window will pop up displaying the
version number.

To enter the application, first of all create
a directory called !FirstFew, and open it up
by double-clicking on it while holding down
Shift. Next use Edit to create an Obey file
called !Run containing the following three
lines, and save it inside the !FirstFew
directory:

W impSlot -min 16K -max 16K
Run <Obey$Dir>.!RunImage %*0

Next type in the listing given and save it
inside !FirstFew under the name !RunImage.
Finally use Paint to design a sprite called
!firstfew (mode 12, 34 pixels wide by 17
pixels high) and save this file as !Sprites in
the !FirstFew directory.

USING THE PROGRAM
Running the application will install an

icon on the right-hand side of the icon bar.
Dragging any file out of a directory viewer
to this icon will open a window displaying
the first 217 bytes of the file as ASCII text
(unprintable characters are displayed as full
stops). Dragging a directory results in an
error box being displayed.

HOW IT WORKS
As the initialisation and polling loop are

in no way peculiar I will only explain the
reading procedure. When a DataLoad
message is received from the Filer it is first
checked for the filetype. Filetypes &1000
and &2000 (directories and applications)
will cause an error to be reported. If the file
is of any other type, reading is attempted.
Should reading fail (e.g. the Filer did not get
the user to insert the right disc, the file has
no read access set etc.) the procedure is
aborted. Only when the file has been read
correctly will FirstFew acknowledge the
DataLoad message and update the display
window. 

10 REM           >!RunImage
20 REM Program   Show First Few Bytes
30 REM Version   A 1.01
40 REM Author    Oliver Schön
50 REM Risc User June 1992
60 REM Program   Subject to Copyright
70 REM           Not Public Domain
80 :
90 ON ERROR PROCerror(ERR,REPORT$+"("

+STR$ERR+") in "+STR$ERL)
100 PROCinit
110 REPEAT
120  PROCpoll
130 UNTIL quit%
140 PROCquit
150 END
160 :
170 DEF PROCinit
180  SYS "Hourglass_On"
190  DIM mblk% &40,iblk% &30,wblk% &60
200  DIM tblk% &100,ublk% &200,eblk% &
40
210  SYS "Wimp_Initialise",200,&4B5341
54,"Version Identifier"
220  PROCinstallicon
230  PROCinitwindow
240  SYS "Hourglass_Off"
250  quit% = FALSE
260 ENDPROC
270 :
280 DEF PROCinstallicon
290  iblk%!0=-1
300  iblk%!4=0: iblk%!8=0
310  iblk%!12=68: iblk%!16=68
320  iblk%!20=&1700300A
330  $(iblk%+24)="!firstfew"
340  SYS "Wimp_CreateIcon",,iblk% TO i
h %
350 ENDPROC
360 :

370 DEF PROCinitwindow
380  LOCAL A%,pt%
390  wblk%!4=300:wblk%!8=200
400  wblk%!12=840:wblk%!16=600
410  wblk%!20=0:wblk%!24=0
420  wblk%!28=-1:wblk%!32=&FF000012
430  wblk%!36=&00070207:wblk%!40=&000C
0103
440  wblk%!44=0:wblk%!48=-400:wblk%!52
=540:wblk%!56=0
450  wblk%!60=&3D:wblk%!64=0:wblk%!68=
1:wblk%!72=0
460  $(wblk%+76)="First Few":wblk%!88=
0
470  SYS "Wimp_CreateWindow",,wblk%+4 
TO !wblk%
480  !ublk%=!wblk%:ublk%!4=8:ublk%!12=
532
490  ublk%!20=&07000139:ublk%!28=-1:ub
lk%!32=32
500  FOR A%=0 TO 7
510   ublk%!8=-(A%*48+56):ublk%!16=-(A
%*48+8):ublk%!24=tblk%+A%*32
520   SYS "Wimp_CreateIcon",,ublk%
530  NEXT
540  $tblk%=STRING$(255," ")
550 ENDPROC
560 :
570 DEF PROCpoll
580  LOCAL reason,A%
590  SYS "Wimp_Poll",&81833,ublk% TO r
eason
600  CASE reason OF
610   WHEN 2: SYS "Wimp_OpenWindow",,u
blk%
620    FOR A%=0 TO 28 STEP 4 : wblk%!A
%=ublk%!A% : NEXT
630   WHEN 3: SYS "Wimp_CloseWindow",,
ublk%
640   WHEN 6: PROChandle_button(ublk%)
650   WHEN 9: quit% = TRUE
660   WHEN 17,18: PROCreceive_mess(ubl



k%)
670  ENDCASE
680 ENDPROC
690 :
700 DEF PROChandle_button(ublk%)
710  CASE (ublk%!8) OF
720   WHEN 2:IF ublk%!12=-2 THEN PROCh
andle_menu(ublk%)
730   WHEN 4:PROCshowdata
740  ENDCASE
750 ENDPROC
760 :
770 DEF PROChandle_menu(ublk%)
780  $mblk%="FirstFew"
790  mblk%!12=&00070207
800  mblk%!16=124:mblk%!20=44:mblk%!24
=0
810  mblk%!28=128 : mblk%!32 = -1
820  mblk%!36=&07009039 : $(mblk%+40)=
"Quit"
830  SYS "Wimp_CreateMenu",,mblk%,!ubl
k%-66,140
840 ENDPROC
850 :
860 DEF PROCshowdata
870  SYS "Wimp_CloseWindow",,wblk%
880  SYS "Wimp_GetWindowState",,wblk%
890  SYS "Wimp_OpenWindow",,wblk%
900 ENDPROC
910 :
920 DEF PROCreceive_mess(ublk%)
930  CASE (ublk%!16) OF
940   WHEN 0: quit% = TRUE
950   WHEN 3: PROCread(ublk%)
960   WHEN &502: IF ublk%!32=-2 THEN P
ROCanswer_help
970  ENDCASE
980 ENDPROC
990 :
1000 DEF PROCread(ublk%)
1010  LOCAL num%,B$,do%,A%,C%,D%,F$
1020  LOCAL ERROR

1030  ON ERROR LOCAL ENDPROC
1040  A%=ublk%+44
1050  WHILE ?A%>31
1060   F$+=CHR$ ?A% : A%+=1
1070  ENDWHILE
1080  IF ublk%!40=&1000 OR ublk%!40=&20
00 THEN
1090   PROCreport(100,"Cannot view dire
ctories.",1)
1100  ELSE
1110   ublk%!12=ublk%!8 : ublk%!16=4
1120   num%=OPENIN F$
1130   C%=tblk%+32 : D%=0
1140   $tblk%=STRING$(255," ")
1150   FOR A%=1 TO 7
1160    SYS "OS_GBPB",3,num%,C%,31,D% T
O ,,C%,,D%
1170    ?C%=13 : C%+=1
1180   NEXT
1190   FOR A%=32 TO 254
1200    IF (tblk%?A%<32) AND ((A%+1)MOD
32 <>0) THEN tblk%?A%=46
1210   NEXT
1220   $tblk%="Read "+FNname(F$)
1230   CLOSE# num%
1240   SYS "Wimp_SendMessage",17,ublk%,
ublk%!4
1250   PROCshowdata
1260  ENDIF
1270 ENDPROC
1280 :
1290 DEF FNname(F$)
1300  LOCAL B$
1310  B$=RIGHT$(F$,13)
1320  WHILE INSTR(B$,".")
1330   B$=RIGHT$(B$,LEN B$ - INSTR(B$,"
."))
1340  ENDWHILE
1350 =B$
1360 :
1370 DEF PROCanswer_help
1380  !ublk%=&D0



1390  ublk%!12=ublk%!8
1400  ublk%!16=&503
1410  $(ublk%+20)="This is the First Fe
w Utility.|M Drop files (no directories)
here to view their first few bytes.|M C
lick MENU to choose Quit."+CHR$(0)
1420  SYS "Wimp_SendMessage",17,ublk%,u
blk%!4
1430 ENDPROC
1440 :
1450 DEF PROCquit
1460  SYS "Wimp_DeleteWindow",,wblk%
1470  iblk%!4=ih%
1480  SYS "Wimp_DeleteIcon",,iblk%
1490  SYS "Wimp_CloseDown"
1500 ENDPROC
1510 :
1520 DEF PROCerror(!eblk%,$(eblk%+4))
1530  SYS "Wimp_ReportError",eblk%,1,"F
irst Few"
1540  SYS "Wimp_CloseDown"
1550 END
1560 :
1570 DEF PROCreport(!eblk%,$(eblk%+4),f
lags%)
1580  SYS "Wimp_ReportError",eblk%,flag
s%,"First Few" TO ,!eblk%
1590 ENDPROC


